
Paste the Full VBA Macro 
' ============================================================= 
' OfficeSkillset — ChatGPT + Excel VBA Integration 
' Compatible: Microsoft 365, Excel 2021, Excel 2019 (Windows) 
' Mac users: see note in the Mac Compatibility section below 
' ============================================================= 
 
Option Explicit 
 
' --------------------------------------------------------------- 
' CONFIGURATION — Replace with your actual OpenAI API key 
' --------------------------------------------------------------- 
Private Const OPENAI_API_KEY As String = "YOUR_API_KEY_HERE" 
Private Const OPENAI_MODEL As String = "gpt-4o-mini" 
Private Const OPENAI_URL As String = "https://api.openai.com/v1/chat/completions" 
Private Const MAX_TOKENS As Long = 500 
Private Const TEMPERATURE As Double = 0.3 
 
' --------------------------------------------------------------- 
' UDF: Use in any Excel cell like =AskChatGPT("Your prompt here") 
' --------------------------------------------------------------- 
Public Function AskChatGPT(prompt As String) As String 
On Error GoTo ErrHandler 
 
Dim http As Object 
Dim jsonBody As String 
Dim response As String 
Dim parsed As String 
 
' Escape special characters in prompt for valid JSON 
Dim safePrompt As String 
safePrompt = Replace(prompt, "\", "\\") 
safePrompt = Replace(safePrompt, """", "\""") 
safePrompt = Replace(safePrompt, Chr(10), "\n") 
safePrompt = Replace(safePrompt, Chr(13), "\r") 
 
' Build JSON request body 
jsonBody = "{""model"":""" & OPENAI_MODEL & """," & _ 
"""messages"":[{""role"":""user"",""content"":""" & safePrompt & """}]," & _ 
"""max_tokens"":" & MAX_TOKENS & "," & _ 
"""temperature"":" & TEMPERATURE & "}" 
 



' Create HTTP object (Windows — MSXML2) 
Set http = CreateObject("MSXML2.XMLHTTP") 
http.Open "POST", OPENAI_URL, False 
http.setRequestHeader "Content-Type", "application/json" 
http.setRequestHeader "Authorization", "Bearer " & OPENAI_API_KEY 
http.Send jsonBody 
 
' Get raw response text 
response = http.responseText 
 
' Parse the "content" field from JSON response 
parsed = ExtractJsonValue(response, """content"":""", """}") 
 
' Clean up escape sequences from the response 
parsed = Replace(parsed, "\n", Chr(10)) 
parsed = Replace(parsed, "\r", "") 
parsed = Replace(parsed, "\""", Chr(34)) 
 
AskChatGPT = parsed 
Exit Function 
 
ErrHandler: 
AskChatGPT = "ERROR: " & Err.Description 
End Function 
 
 
' --------------------------------------------------------------- 
' BATCH SUB: Run ChatGPT on a column of prompts (Column A → Column B) 
' Usage: Select any cell, then run this Sub from Macros menu 
' --------------------------------------------------------------- 
Public Sub BatchAskChatGPT() 
Dim ws As Worksheet 
Dim lastRow As Long 
Dim i As Long 
Dim promptText As String 
Dim result As String 
 
Set ws = ActiveSheet 
lastRow = ws.Cells(ws.Rows.Count, "A").End(xlUp).Row 
 
Application.ScreenUpdating = False 
Application.Calculation = xlCalculationManual 
 
For i = 2 To lastRow ' Assumes Row 1 = Headers 
promptText = ws.Cells(i, 1).Value 
If Len(Trim(promptText)) > 0 Then 
result = AskChatGPT(promptText) 
ws.Cells(i, 2).Value = result 
End If 
DoEvents ' Allow Excel to breathe during long loops 
Next i 



 
Application.ScreenUpdating = True 
Application.Calculation = xlCalculationAutomatic 
 
MsgBox "Batch complete! " & (lastRow - 1) & " rows processed.", vbInformation 
End Sub 
 
 
' --------------------------------------------------------------- 
' HELPER: Minimal JSON string value extractor 
' Extracts text between a known key marker and end quote 
' --------------------------------------------------------------- 
Private Function ExtractJsonValue(jsonStr As String, _ 
startMarker As String, _ 
endMarker As String) As String 
Dim startPos As Long 
Dim endPos As Long 
Dim extracted As String 
 
startPos = InStr(jsonStr, startMarker) 
If startPos = 0 Then 
ExtractJsonValue = "PARSE_ERROR: key not found" 
Exit Function 
End If 
 
startPos = startPos + Len(startMarker) 
endPos = InStr(startPos, jsonStr, endMarker) 
 
If endPos = 0 Then 
ExtractJsonValue = "PARSE_ERROR: end marker not found" 
Exit Function 
End If 
 
extracted = Mid(jsonStr, startPos, endPos - startPos) 
ExtractJsonValue = extracted 
End Function 
 
AI Prompt Section 
 
These prompts are designed to be used directly inside the AskChatGPT() function. Copy, paste 
into a cell, and adjust the cell references to match your data. 
 
Prompt 1 — Data Classification 
=AskChatGPT("Classify this customer feedback as Positive, Negative, or Neutral. Only reply 
with one word. Feedback: "&A2) 
Prompt 2 — Summary Generation 
=AskChatGPT("Summarize the following in exactly 2 sentences: "&A2) 
Prompt 3 — Email Draft from Data Row 
=AskChatGPT("Write a professional follow-up email to "&A2&" about their order #"&B2&". Keep 
it under 80 words.") 
Prompt 4 — Formula Explanation 



=AskChatGPT("Explain this Excel formula to a non-technical person in 1 sentence: "&A2) 
Prompt 5 — Product Description Generator 
=AskChatGPT("Write a 30-word product description for an e-commerce listing. Product: "&A2&". 
Key features: "&B2) 
Prompt 6 — Language Translation 
=AskChatGPT("Translate the following text to Spanish. Return only the translation: "&A2) 
Prompt 7 — Data Cleaning Instruction 
=AskChatGPT("Fix the spelling and grammar in this text. Return only the corrected version: 
"&A2) 
Prompt 8 — Tone Rewrite 
=AskChatGPT("Rewrite the following in a friendly, casual tone suitable for social media: "&A2) 

